IDENTIFICATION

Product Code:	MAINDEC-08-D2PE-D
Product Name:	Family-of-8 ASR 33/35 Teletype Tests Part 1
Date Created:	February 21, 1969
Maintainer:	Diagnostic Group

1. ABSTRACT

The Family-of-8 ASR33/35 Teletype Tests Part 1 is the first part of a two part package used to test the ASR33, ASR33TY, or ASR35 Teletype when attached to a Family-of-8 system.

Part 1 contains nine selectable programs numbered from 0 to 10 (octal). The programs are selected by means of the switch register (SR).

The programs available are:

PRG0	Basic Input Logic Tests
PRG1	Basic Output Logic Tests
PRG2	Reader Test
PRG3	Test Tape Generator. Punches tape with characters stored in locations 0021 and 0022.
PRG4	Test Tape Generator. Punches Binary Count Pattern test tape.
PRG5	Reader Exerciser. Reads Binary Count pattern tape in random length blocks, and with fixed stalls between characters. The stall is determined at random.
PRG6	Reader Exerciser. Reads Binary Count pattern tape. Fixed stall between characters. Stall count is taken from LOC 0023.
PRG7	Reader Exerciser. Reads tape punched with any 2 test characters. Random length blocks and fixed stall between characters. The stall is determined at random.
PRG10	Reader Exerciser. Reads tape punched with any 2 test characters. Fixed stall between characters. Stall count taken from LOC 0023.
PRG11	ASR33TY Automatic Reader option test. Checks for correct response to READER ON, and READER OFF commands.
PRG12	ASR33TY Automatic Punch option test. Checks for correct response to PUNCH ON and PUNCH OFF commands

2. REQUIREMENTS

2.1 Equipment

Standard PDP-8/S, PDP-8 or PDP-8/I with ASR33, ASR33TY, or ASR35 Teletype.

NOTE

Programs PRG0 through PRG10 are not written specifically for the ASR33TY Teletype. Whenever these programs are run, be sure to lock the punch and reader in their ON position. In the case of the reader, use a heavy rubber band to keep the reader switch in the ON position.

2.2 Storage

Locations 0000 through 2341 are used.

3. LOADING PROCEDURE

3.1 Method

The Binary Loader is used to load the program.

4. STARTING PROCEDURE (PRG0)

4.1 Control Switch Settings (PRG0)

SR0	Halt at end of routine. Routine number in AC.
SR1	Select routine whose number is set in SR6 through SR11.
SR2	Loop program.
SR6 through SR11	Routine number to be selected.

4.2 Starting Addresses (PRG0)

This program starts at LOC 0200.

4.3 Program and/or Operator Action (PRG0)

a. Insure Teletype is on-line.

b. Load binary count pattern test tape in reader.

c. Turn on reader.

d. Load address 0200.

e. Set SR to 0000.

f. Press START.

g. Program halts at LOC 0232 to permit setting of options.

h. Select desired options, if any, in SR. For normal run SR should be 0000. Press CONTINUE.

i. Program is executed and halts at LOC 0274, program end halt, if no loop options are selected and if no errors occur.

5. OPERATING PROCEDURE (PRG0)

5.1 Program and/or Operator Action (PRG0)

5.1.1 Normal Halts (PRG0)

LOC 0232	SR SET halt. Occurs to permit setting of desired options.
LOC 0274	Program end halt. Occurs if no "loop program" option is set. Set desired options and press CONTINUE. If no options are set, this halt reoccurs.
LOC 0320	Routine end halt. Occurs at end of routine if SR0 = 1. To proceed, press CONTINUE.

6. ERRORS (PRG0)

6.1 Error Halts and Description (PRG0)

LOC 0177	Incorrect program number selected. Set SR to 0000 and press CONTINUE.
LOC 0255	Nonexistent routine selected. Set correct routine number in SR6 through SR11 and press CONTINUE.
LOC 1221	PRG0, routine 0 error halt. KCC instruction failed to clear the AC. Pressing CONTINUE enters scope loop that sets AC to all 1s, issues KCC to clear AC, and repeats. Manual restart.
LOC 1244	PRG0, routine 1 error halt. 200 ms after KRB instruction the flag is not set, or KSF instruction failed to skip on flag = 1. Pressing CONTINUE repeats the test.
LOC 1267	PRG0, routine 2 error halt A. Same as PRG0, routine 1 error halt.
LOC 1271	PRG0, routine 2 error halt B. KSF instruction failed to skip with flag = 1. Pressing CONTINUE enters scope loop that skips on flag continuously. Manual Restart.
LOC 1320	PRG0, routine 3 error halt A. Same as PRG0, routine 1 error halt.
LOC 1322	PRG0, routine 3 error halt B. KCC failed to reset, or KSF instruction skipped with flag = 0. Pressing CONTINUE enters scope loop that clears the flag and skips on flag continuously. Manual restart.
LOC 1345	PRG0, routine 4, error halt A. Unexpected interrupt. Turn off any device that may be causing an interrupt (other than the Teletype). Pressing CONTINUE repeats the test.
LOC 1362	PRG0, routine 4, error halt B. With reader flag =1 and interrupt enabled, no interrupt occurred. Pressing CONTINUE enters scope loop that turns on interrupt continuously. Manual restart.
LOC 1417	PRG0, routine 5, error halt. Timing error. Flag not -1 110 ms after KRB command. Pressing CONTINUE enters scope loop that reads tape continuously to aid in timing adjustment. Manual restart.

LOC 1457	PRG0, routine 6, error halt A. Reread error. A reread of the Teletype buffer did not match with the original read. New character is displayed in AC. Press CONTINUE.
LOC 1462	PRG0, routine 6, error halt B. Follow up halt to PRG0, routine 6, error halt A. The "old" character is displayed in AC. Pressing CONTINUE enters scope loop that reads the teletype buffer continuously. Manual restart.

4A. STARTING PROCEDURES (PRG1)

4.1A Control Switch Settings (PRG1)

SR0	Halt at end of routine. Routine number in AC.
SR1	Select routine whose number is set in SR6 through SR11.
SR2	Loop program.
SR6 through SR11	Routine number to be selected.

4.2A Starting Addresses (PRG1)

This program starts at LOC 0200.

4.3A Program and/or Operator Action (PRG1)

a. Insure Teletype is on-line.

b. Insure reader is off.

c. Insure that there is paper in teleprinter.

d. Load address 0200.

e. Set SR to 0001.

f. Press START.

g. Program halts at LOC 0232 to permit setting of options.

h. Select desired options, if any, in SR. For normal run SR should be 0000. Press CONTINUE.

i. Program is executed and halts at LOC 0274, program end halt, if no loop options are selected and if no errors occur.

5.A OPERATING PROCEDURE (PRG1)

5.1A Program and/or Operator Action (PRG1)

5.1.1A Normal Halts (PRG1)

LOC 0232	SR SET halt. Occurs to permit setting of desired options. Press CONTINUE.
LOC 0274	Program end halt. Occurs if no "loop program" option is set. Set desired options and press CONTINUE. If no options are set, the halt reoccurs.
LOC 0320	Routine end halt. Occurs at end of routine if SR0 = 1. To proceed, press CONTINUE.

6.A ERRORS (PRG1)

6.1A Error Halts and Description (PRG1)

LOC 0177	Incorrect program number selected. Set SR to 0001 and press CONTINUE.
LOC 0255	Nonexistent routine selected. Set correct routine number in SR6 through SR11 and press CONTINUE.
LOC 1627	PRG1, routine 0, error halt A. 200 ms after TLS command the flag is not 1, or TSF command failed to skip. Pressing CONTINUE repeats the test.
LOC 1631	PRG1, routine 0, error halt B. With flag = 1, TSF command failed to skip. Pressing CONTINUE enters scope loop that skips on flag continuously. Manual restart.
LOC 1651	PRG1, routine 1 error halt. TCF command failed to clear flag, or TSF command skipped with flag = 0. Pressing CONTINUE enters scope loop that clears the flag and then skips on flag continuously. Manual restart.
LOC 1676	PRG1, routine 2 error halt. TCF command failed to clear flag. Pressing CONTINUE enters scope loop that issues TCF command continuously. Manual restart.
LOC 1717	PRG1, routine 3, error halt A. Unexpected interrupt. Turn off any device that may be causing an interrupt. (The teletype reader must be off). Press CONTINUE to repeat test.
LOC 1734	PRG1, routine 3, error halt B. With flag = 1, and interrupt enabled, no interrupt occurred. Pressing CONTINUE enters scope loop that turns on interrupt continuously. Manual restart.
LOC 1765	PRG1, routine 4 error halt. Timing error. Flag not 1 110 ms after TLS command. Pressing CONTINUE enters scope loop that runs the printer/punch continuously, to aid in timing adjustment. Manual restart.

4.B STARTING PROCEDURES (PRG2)

4.1B Control Switch Settings (PRG2)

SR0	Halt at end of routine. Routine number in AC.
SR1	Select routine whose number is set in SR6 through SR11.
SR2	Loop program.
SR6 through SR11	Routine number to be selected.

4.2B Starting Addresses (PRG2)

This program starts ar LOC 0200.

4.3B Program and/or Operator Action (PRG2)

a. Insure Teletype is on-line.

b. Load binary count pattern test tape in reader.

c. Turn on reader.

d. Load address 0200.

e. Set SR to 0002.

f. Press START

g. Program halts at LOC 0232 to permit setting of options.

h. Set desired options, if any, in SR. For normal run, SR should be 0000. Press CONTINUE.

i. Program is executed and halts at LOC 0274, program end halt, if no loop options are set, and if no errors occur.

5.B OPERATING PROCEDURE (PRG2)

5.1B Program and/or Operator Action (PRG2)

5.1.1B Normal Halts (PRG2)

LOC 0232	SR SET halt. Occurs to permit setting of desired options. Press CONTINUE.
LOC 0274	Program end halt. Occurs if no "loop program" option is set, set options and press CONTINUE. If no options are set, this halt reoccurs.
LOC 0320	Routine end halt. Occurs at end of routine if SR0 = 1. To proceed press CONTINUE.

6.B ERRORS (PRG2)

6.1B Error Halts and Description (PRG2)

LOC 0177	Incorrect program number selected. Set SR to 0002 and press CONTINUE.
LOC 0255	Nonexistent routine selected. Set correct routine number in SR6 through SR11 and press CONTINUE.
LOC 0564	Unable to sync. Sync subroutine has not found an all 1's character within 256 characters. Press CONTINUE to retry.
LOC 2030	PRG2, routine 0, error halt A. Read error. Bad character in AC. Press CONTINUE.
LOC 2033	PRG2, routine 0, error halt B. Follow up halt. Expected character in AC. Pressing CONTINUE resumes test.
LOC 2062	PRG2, routine 1, error halt A. Read error. Bad character in AC. Press CONTINUE.
LOC 2065	PRG2, routine 1, error halt B. Follow up halt. Expected character in AC. Pressing CONTINUE resumes test.
LOC 2120	PRG2, routine 2, error halt A. Read error. Bad character in AC. Press CONTINUE.
LOC 2123	PRG2, routine 2, error halt B. Follow up halt. Expected character AC. Pressing CONTINUE resumes test.

4.C STARTING PROCEDURES (PRG3)

4.1.C Control Switch Settings (PRG3)

None

4.2C Starting Addresses (PRG3)

This program starts at LOC 0200.

4.3C Program and/or Operator Action (PRG3)

a. Insure Teletype is on-line.

b. Turn off teletype reader.

c. Load blank tape in punch.

d. Turn on punch.

e. Deposit in LOC 0021 and 0022 (8), the 8-bit code for characters to be punched.

f. Load address 0200.

g. Set SR to 0003.

h. Press START.

i. Program punches tape until stopped by user.

5.C OPERATING PROCEDURE (PRG3)

5.1C Program and/or Operator Action (PRG3)

5.1.1C Normal Halts (PRG3)

None

6.C ERRORS (PRG3)

6.1C Error Halts and Description (PRG3)

LOC 0177	Incorrect program number selected. Set SR to 0003 and press CONTINUE.

4.D STARTING PROCEDURES (PRG4)

4.1D Control Switch Settings (PRG4)

None

4.2D Starting Addresses (PRG4)

This program starts at LOC 0200.

4.3D Program and/or Operator Action (PRG4)

a. Insure Teletype is on-line.

b. Turn off teletype reader.

c. Load blank tape in punch.

d. Turn on punch.

e. Load address 0200.

f. Set SR to 0004.

g. Press START.

h. Program punches binary count pattern test tape until stopped user.

5.D OPERATING PROCEDURE (PRG4)

5.1D Program and/or Operator Action (PRG4)

5.1.1D Normal Halts (PRG4)

None.

6.D ERRORS (PRG4)

6.1D Error Halts and Description (PRG4)

LOC 0177	Incorrect program number selected. Set SR to 0004 and press CONTINUE.

4.E STARTING PROCEDURES (PRG5)

4.1E Control Switch Settings (PRG5)

SR0	Halt. Program halts with accumulated error count in AC.
SR5	Halt on error. Program halts if read-error occurs.

4.2E Starting Addresses (PRG5)

This program starts at LOC 0200.

4.3E Program and/or Operator Action (PRG5)

a. Insure Teletype is on-line.

b. Load binary count pattern test tape in reader.

c. Turn on reader.

d. Load address 0200.

e. Set SR to 0005.

f. Press START.

g. Program runs continuously until stopped, unless a read error occurs with SR5 = 1, or SR0 is set to 1.

5.E OPERATING PROCEDURE (PRG5)

5.1E Program and/or Operator Action (PRG5)

5.1.1E Normal Halts

LOC 1115	Halt. Accumulated errors in AC. Occurs if SR0 = 1. Press CONTINUE to proceed.

6.E ERRORS (PRG5)

6.1E Error Halts and Description (PRG5)

LOC 0177	Incorrect program number selected. Set SR to 0005 and press CONTINUE.
LOC 1110	Read error halt. Occurs if SR5 = 1. Press CONTINUE to proceed.

4.F STARTING PROCEDURES (PRG6)

4.1F Control Switch Settings (PRG6)

SR0	Halt. Program halts with accumulated error count in AC.
SR5	Halt on error. Program halts if read error occurs.

4.2F Starting Addresses (PRG6)

This program starts at LOC 0200.

4.3F Program and/or Operator Action (PRG6)

a. Insure Teletype is on-line.

b. Load binary count pattern test tape in reader.

c. Turn on reader.

d. Deposit in LOC 0023 the desired stall count in 2's complement form. A count of -1 gives a 1ms stall, etc.

e. Load address 0200.

f. Set SR to 0006.

g. Press START.

h. Program runs continuously until stopped, unless a read error occurs with SR5 = 1, or SR0 is set to 1.

5.F OPERATING PROCEDURE (PRG6)

5.1F Program and/or Operator Action (PRG6)

5.1.1F Normal Halts (PRG6)

LOC 1115	Halt. Accumulated errors in AC. Occurs if SR0 = 1. Press CONTINUE to proceed.

6.F ERRORS (PRG6)

6.1F Error Halts and Description (PRG6)

LOC 0177	Incorrect program number selected. Set SR to 0006 and press CONTINUE.
LOC 1110	Read error halt. Occurs if SR5 = 1. Press CONTINUE to proceed.

4.G STARTING PROCEDURES (PRG7)

4.1G Control Switch Settings (PRG7)

SR0	Halt. Program halts with accumulated error count in AC.
SR5	Halt on error. Program halts if read error occurs.

4.2G Starting Addresses (PRG7)

This program starts at LOC 0200.

4.3G Program and/or Operator Action (PRG7)

a. Insure Teletype is on-line.

b. Load reader with 2-character test tape.

c. Turn on reader.

d. Deposit in location 0021 and 0022 the 8-bit codes for the character punched in the test tape.

e. Load address 0200.

f. Set SR to 0007.

g. Press START

h. Program runs continuously until stopped, unless a read error occurs with SR5 = 1, or SR0 is set to 1.

5.G OPERATING PROCEDURE (PRG7)

5.1G Program and/or Operator Action (PRG7)

5.1.1G Normal Halts (PRG7)

LOC 1115	Halt. Accumulated errors in AC. Occurs if SR0 = 1. Press CONTINUE to proceed.

6.G ERRORS (PRG7)

6.1G Error Halts and Description (PRG7)

LOC 0177	Incorrect program number selected. Set SR to 0007 and press CONTINUE.
LOC 1110	Read error halt. Occurs if SR5=1. Press CONTINUE to proceed.
LOC 1137	Align error halt. Insure that correct tape is used, and check Step 4.3Gd.

4.H STARTING PROCEDURES (PRG 10)

4.1H Control Switch Settings (PRG 10)

SR0	Halt. Program halts with accumulated error count in AC.
SR5	Halt on error. Program halts if read error occurs.

4.2H Starting Addresses (PRG 10)

This program starts at LOC 0200.

4.3H Program and/or Operator Action (PRG 10)

a. Insure Teletype is on-line.

b. Load reader with 2-character test tape.

c. Turn on reader

d. Deposit in LOC 0023 the desired stall count in 2's complement form. A count of -1 gives a 1 ms stall, etc.

e. Load address 0200.

f. Set SR to 0010.

g. Press START.

h. Program runs continuously until stopped, unless a read error occurs with SR5 = 1, or SR0 is set to 1.

5.H OPERATING PROCEDURE (PRG10)

5.1H Normal Halts (PRG 10)

LOC 1115	Halt. Accumulated errors in AC. Occurs is SR0 = 1. Press CONTINUE to proceed.

6.H ERRORS (PRG 10)

6.1H Error Halts and Description (PRG 10)

LOC 0177	Incorrect program number selected. Set SR to 0007 and press CONTINUE.
LOC 1110	Read error halt. Occurs if SR5 = 1. Press CONTINUE to proceed.
LOC 1137	Align error halt. Insure that correct tape is used, and check Step 4.3Hd.

4.I STARTING PROCEDURES (PRG 11)

4.1I Control Switch Settings (PRG 11)

None

4.2I Starting Addresses (PRG 11)

This program starts at LOC 0200.

4.3I Program and/or Operator Action (PRG 11)

a. Insure that Teletype is on-line.

b. Load reader with any test tape loop.

c. Turn on reader by pushing the momentary contact switch to the START position.

d. Make sure that teletype punch is not locked on.

e. Load Address 0200.

f. Set SR to 0011.

g. Press START

h. Program runs continuously until stopped, unless an error halt occurs.

5.I OPERATING PROCEDURE (PRG 11)

5.1I Normal Halts (PRG 11)

None

6.I ERRORS (PRG 11)

6.1I Error halts and Description (PRG 11)

LOC 2212	Reader flag not set after approximately 110 ms after KCC command issued after READER ON command. Probably the READER ON command failed to turn on the reader. Press CONTINUE to proceed.
LOC 2227	Reader flag was set after approximately 110 ms after KCC command issued after READER OFF command. Probably the READER OFF command failed to turn off the reader. Press CONTINUE to proceed.

4.J STARTING PROCEDURES (PRG 12)

4.1J Control Switch Settings (PRG 12)

None

4.2J Starting Addresses (PRG 12)

This program starts at LOC 0200.

4.3J Program and/or Operator Action (PRG 12)

a. With Teletype off-line, punch a section of blank leader about 6 inches long. Return Teletype to on-line position.

b. Load leader on reader, leaving very little slack between punch and reader.

c. Turn on reader by pushing the momemtary contact switch to the START position.

d. Make sure that teletype punch is not locked on.

e. Load address 0200.

f. Set SR to 0012.

g. Press START

h. Program runs continuously until stopped, unless an error halt occurs.

5.J OPERATING PROCEDURE (PRG 12)

5.1J Normal Halts (PRG 12)

None

6.J ERRORS (PRG 12)

6.1J Error halts and Description (PRG 12)

LOC 2337	Reader failed to read a rubout. Reader failed to read correctly if character on tape is a rubout. If tape character is a rubout, the PUNCH FEED OFF command failed to stop the punch from feeding. Check for other similar failures. To be correct, the punched tape should contain all rubouts. Press CONTINUE to proceed.

6.2J Other Errors (PRG 12)

Failure of the PUNCH FEED ON command will eventually be detected by the tightening of the slack between the reader and punch. The longer the program is run the better the chances are of detecting the problem, if present.

7. RESTRICTIONS

7.1 Starting Restrictions

All programs must be started at LOC 0200.

7.2 Operating Restrictions

PRG0 and PRG1 must be run prior to executing any other programs. Problems detected during execution of PRG0 and PRG1 should be corrected as they occur.

PRG11 must precede PRG12 execution.

8. MISCELLANEOUS

8.1 Execution Time

PRG0 execution time: 1 minute

PRG1 execution time: 20 seconds

PRG2 execution time: 18 minutes

PRG3 through PRG12 are continuous run programs.

8.2 Test Tapes

MAINDEC-08-D2G3-PT Binary Count Pattern test tape is provided with this program. For convenience in use, the tape should be spliced into a loop, making sure that the pattern is matched at the splice point.

9. PROGRAM DESCRIPTION

The Family-of-8 ASR33/35 Teletype Tests, Part 1, consists of 11 programs numbered from 0 to 12 (octal).

9.1 PRG0 - Basic Input Logic Tests

This program contains 7 routines numbered from 0 to 6 (octal).

RTN0	Checks that KCC command is abie to clear the AC. Test is done 1000 times.
RTN1	Issues KCC, waits 200 ms and checks for flag = 1. A failure to skip on flag indicates that flag is not 1, or KSF command failure to skip.

RTN2 Checks ability of KSF command to skip with flag = 1. Done 1000 times.

RTN3 Checks that KSF command does not skip with flag = 0. Done 500 times.

RTN4 Checks that no other device can cause an interrupt, and then checks that the reader is capable of interrupting.

RTN5 Timing Test.

RTN6 Reads a character from tape and saves it. It then rereads the TTI statically 1000 times to check for consistent reading from TTI. 256 characters are read in this manner.

9.2 PRG1 - Basic Output Logic Tests

This program contains five routines numbered from 0 to 4.

RTN0 Issues TLS, waits 200 ms, and checks for flag = 1. A failure to skip indicates that flag is not 1, or KSF command failed. If this part is satisfied the routine skips on flag = 1, 1000 times. Failure to skip indicates TSF failure.

RTN1 Checks that TSF command does not skip with flag = 0. Done 1000 times.

RTN2 Checks that TCF command clears flag. Done 100 times.

RTN3 Checks that no other device can cause an interrupt, and then checks that the printer/punch is able to interrupt.

RTN4 Timing Test.

9.3 PRG2 - Reader Test

This program contains three routines numbered from 0 to 2.

RTN0 Reads 4095 characters of binary count pattern, at full speed.

RTN1 Reads 2000 characters of binary count patterns with random stalls between characters.

RTN2 Reads 100 random-length character blocks. Fixed stall between characters in a block. Stall is changed for each block and is determined at random.

9.4 PRG3 - Test Tape Generator

This program punches test tape with characters whose code is stored in LOC 0021 and 0022.

9.5 PRG4 - Test Tape Generator

Punches binary count pattern test tape.

9.6 PRG5 - Reader Exerciser

This program reads binary count pattern test tape, in random length blocks, and with fixed stalls between characters. Stall is determined at random.

9.7 PRG6 - Reader Exerciser

Reads binary count pattern test tape. Fixed stall between characters. Stall count is taken from LOC 0023.

9.8 PRG7 - Reader Exerciser

Reads test tape punched with any two test characters, random length blocks, and fixed stall between characters. Stall is determined at random.

9.9 PRG10 - Reader Exerciser

Reads test tape punched with any two test characters. Fixed stall between characters. Stall count taken from LOC 0023.

9.10 PRG11 - ASR33TY Automatic Reader Option Test

Checks for correct response to READER ON, and READER OFF commands by checking for correct state of reader flag 110 ms after issuing KCC command which is preceded by one of the reader control commands.

The coder control commands used are:

READER ON - 221

READER OFF - 223

9.11 PRG12 - ASR33TY Automatic Punch Option Test

Checks for correct operation of PUNCH FEED ON and PUNCH FEED OFF commands by punching rubouts with the punch feed on, and all 0's characters with the punch feed off. The resulting tape should contain all rubouts. The tape is verified by running it through the reader at the same time.

The punch control commands used are:

PUNCH FEED ON - 222

PUNCH FEED OFF - 224

```
/FAMILY OF ASR33/35 TELETYPE TESTS - PART 1
/
/COPYRIGHT 1969, DIGITAL EQUIPMENT CORPORATION, MAYNARD, MASS.
/
/PRG0-BASIC INPUT CONTROL LOGIC TEST - (USES READER)
/PRG1-BASIC OUTPUT CONTROL LOGIC TEST - (USES PRINTER)
/PRG2-READER TEST
/PRG3-TEST TAPE GENERATOR, PUNCHES CONTENTS OF LOC 0021 AND 0022
/PRG4-TEST TAPE GENERATOR, PUNCHES BINARY COUNT PATTERN TEST TAPE
/PRG5-READER EXERCISER, READS BINARY COUNT PATTERN TEST TAPE IN RANDOM
/       LENGTH BLOCKS, STALLS WITH FIXED DELAY BETWEEN CHARACTERS. STALL
/       IS DIFFERENT FOR EACH BLOCK.
/PRG6-READER EXERCISER READS BINARY COUNT PATTERN, FIXED STALL BETWEEN CHARACTERS.
/PRG7-READER EXERCISER, READS TAPE PUNCHED BY PRG3, TEST DATA MUST BE STORED
/       IN LOC 0021 AND 0022, RANDOM LENGTH BLOCKS, FIXED STALL BETWEEN
/       CHARACTERS, STALL DIFFERENT FOR EACH BLOCK
/PRG10-SAME AS PRG7, BUT FIXED STALL BETWEEN CHARACTERS (NO RANDOM LENGTH BLOCKS)
/PRG11-ASR33TY AUTOMATIC READER OPTION TEST
/PRG12-ASR33TY AUTOMATIC PUNCH OPTION TEST
/
/SR OPTIONS
/
/SR0-HALT AT END OF ROUTINE, ROUTINE NUMBER IN AC
/SR1-SELECT ROUTINE WHOSE NUMBER IS SET IN SR6 TO SR11
/SR2-LOOP PROGRAM
/SR5-HALT ON ERROR
/SR6-SR11-ROUTINE NUMBER TO BE SELECTED.
```

```
                /FAMILY-OF-8 ASR33/35 TELETYPE TESTS-PART 1
      0000      *0
0000  0000              0000
0001  5001              JMP 1
0002  0002              2
0003  0003              3
      0005      *5
0005  5402              JMP I 2
0006  0000              0
      0020      *20
0020  0000      KSTART, 0                       /USER PROGRAM START,
0021  0000      PTEMP,  0
0022  0000      PTEMP1, 0
0023  0000      DELAYM, 0
0024  0257      CHAIN,  CHAINN                  /CHAIN RTN ENTRY,
0025  0313      SHLT,   SHALT                   /HALT TEST ENTRY
0026  0322      SETCTR, STCTR                   /SET COUNTER ENTRY
0027  0333      DLY1SC, DLYSC                   /DELAY SECONDS ENTRY
0030  0345      DLY1MS, DLYMS                   /DELAY MILLISECS  ENTRY
0031  0232      SRST,   SRSET
0032  0400      RANDNO, RANGEN
0033  0017      PRGMSK, 17
0034  7766      PRGLIM, -12
0035  0000      PRGNUM, 0
0036  0037      PSW,    PRGTAB
0037  1200      PRGTAB, PRG0
0040  1600              PRG1
0041  2000              PRG2
0042  1000              PRG3
0043  1006              PRG4
0044  1013              PRG5
0045  1031              PRG6
0046  1042              PRG7
0047  1060              PRG10
0050  2200              PRG11
0051  2244              PRG12
```

```
0052  0000  TEMP,   0                /WORK
0053  0000  TEMP1,  0                /LOCATIONS
0054  0077  TSIMSK, 77               /SR 6-11 ENABLE MASK
0055  0100  SR5MSK, 100              /SR5 MASK
0056  0000  CPID,   0                /IDENTIFIES CPU
0057  0000  CURTST, 0                /FOR CURRENT TEST ADDRESS
0060  0000  RTNNO,  0                /FOR CURRENT TEST NUMBER
0061  0000  NXTST,  0                /FOR NEXT TEST ADDRESS
0062  0000  SCCTR,  0                /SECONDS COUNTER
0063  0000  MSCTR,  0                /MILLISECONDS COUNTER
0064  0000  MILCTR, 0
0065  0000  MIL1,   0                /FOR 1 MSEC CONSTANT
0066  7444  KP8,    -334             /PDP8 1 MSEC CONSTANT
0067  7764  KP8S,   -14              /PDP8S 1 MSEC CONSTANT
0070  0000  TEMQ,   0                /CONSTANTS
0071  0000  TEMR,   0                /FOR
0072  0000  FLAG,   0                /TYPE
0073  0077  K77,    77               /CHARACTER
0074  7740  M40,    -40              /STRING
0075  0100  C100,   100              /SUBROUTINE
0076  0240  C240,   240
0077  7500  SKIPMA, SMA
0100  7510  SKIPPA, SPA
0101  0000  CTRA,   0                /COUNTER A,
0102  0000  CTRB,   0                /COUNTER B,
0103  0000  SCNT,   0
0104  7634  K100,   -144
0105  2000  K2000,  2000
              /
0106  0546  SYNC,   SYNK             /ENTRY TO SYNC TAPE RTN,
0107  0444  INPATT, INITPT           /ENTRY TO INITIATE PATTERN
0110  0453  GETPT,  GETPTT          /ENTRY TO GET PATTERN CHAR,
0111  0531  CHECK,  CHCK
0112  0502  CRCNT,  CHRCNT
0113  0520  DLYCNT, DLCNT
0114  0000  PFLAG,  0
0115  0465  UPUNCH, PUNCH
0116  0600  UMOVE,  MOVE
0117  7401  MRBOUT, -377
```

```
                    /CONTROL ROUTINE
          0177      * 177
0177      7402              HLT                 /INCORRECT PROGRAM NUMBER
0200      7604      START,  LAS
0201      0033              AND PRGMSK
0202      1034              TAD PRGLIM
0203      7540              SMA SZA
0204      5177              JMP 177
0205      7604              LAS
0206      0033              AND PRGMSK
0207      3035              DCA PRGNUM
0210      1035              TAD PRGNUM
0211      1036              TAD PSW
0212      3052              DCA TEMP
0213      1452              TAD I TEMP
0214      3231              DCA PRGADR
0215      7350      ID,     CLA CLL CMA RAR
0216      7710              SPA CLA
0217      5222              JMP .+3
0220      1066              TAD KP8
0221      7410              SKP
0222      1067              TAD KP8S
0223      3065              DCA MIL1
0224      4516              JMS I UMOVE         /INITIALIZE
0225      0005              5                   /INTERRUPT,
0226      0001              1                   /AREA,
0227      7776              -2
0230      5631              JMP I .+1
0231      0000      PRGADR, 0
0232      7602      SRSET,  HLT CLA

0233      7200      GETRDY, CLA
0234      1020              TAD KSTART          /SET ADDRESS OF 1ST ROUTINE
0235      3061              DCA NXTST           /STORE AT NXTST
0236      4276              JMS FORWD
0237      7604              LAS                 /READ SR
0240      7004              RAL
0241      7500              SMA                 /ROUTINE SELECT?
0242      5457              JMP I CURTST        /NO,  START WITH 1ST RTN
0243      7604              LAS                 /YES
0244      0054              AND TSTMSK
0245      7041              CIA
0246      1060              TAD RTNNO
0247      7650              SNA CLA             /IS IT THIS RTN?
0250      5457              JMP I CURTST        /YES,  GO DO IT
0251      1061              TAD NXTST           /NO
0252      7001              IAC                 /IS THIS LAST TRN?
0253      7640              SZA CLA             /NO
0254      5236              JMP GETRDY+3
0255      7402      INCRTN, HLT                 /YES,  INCORRECT ROUTINE NO,
0256      5233              JMP GETRDY
```

```
0257  4313  CHAINN, JMS SHALT        /HALT? (SRØ)
0252  7604          LAS              /READ SR
0251  7006          RTL
0252  7630          SZL CLA          /SELECT ROUTINE? (SR1)
0253  5233          JMP GETRDY       /YES
0254  1061          TAD NXTST
0255  7001          IAC
0256  7640          SZA CLA          /LAST ROUTINE?
0257  5236          JMP GETRDY+3     /NO,
0270  7604          LAS
0271  7006          RTL
0272  7710          SPA CLA          /LOOP PROGRAM? (SR2)
0273  5233          JMP GETRDY       /YES
0274  7402  PRGEND, HLT              /END OF PROGRAM HALT
0275  5257          JMP CHAINN
0276  0000  FORWD,  Ø
0277  7300          CLA CLL
0300  1461          TAD I NXTST      /GET NEXT RTN NO
0301  3060          DCA RTNNO        /STORE AT RTNNO
0302  2061          ISZ NXTST
0303  1061          TAD NXTST        /SET CURRENT
0304  3052          DCA TEMP         /RTN NUMBER
0305  2061          ISZ NXTST
0306  1061          TAD NXTST        /SET CURRENT
0307  3057          DCA CURTST       /RTN ADDR,
0310  1452          TAD I TEMP       /SET NEXT
0311  3061          DCA NXTST        /RTN ADDR,
0312  5676          JMP I FORWD      /EXIT
```

```
0313  0000  SHALT,  0
0314  7604          LAS             /READ SR
0315  7700          SMA CLA         /HALT? (SR0)
0316  5713          JMP I SHALT
0317  1060          TAD RTNND
0320  7402          HLT             /UNCONDITIONAL HALT (SR0 = 1)
0321  5713          JMP I SHALT     /EXIT,S/-10L

0322  0000  STCTR,  0
0323  7200          CLA
0324  1722          TAD I STCTR     /GET CTR ADDR
0325  3052          DCA TEMP        /AND SAVE AT TEMP
0326  2322          ISZ STCTR
0327  1722          TAD I STCTR     /GET COUNT AND
0330  3452          DCA I TEMP      /STORE PER C(TEMP)
0331  2322          ISZ STCTR
0332  5722          JMP I STCTR     /EXIT
0333  0000  DLYSC,  0
0334  7300          CLA CLL
0335  1733          TAD I DLYSC     /GET SECONDS COUNT
0336  3062          DCA SCCTR       /STORE AT SCCTR
0337  4345          JMS DLYMS               /GO DELAY
0340  6030          -1750           /1 SECOND (1000 MSEC),
0341  2062          ISZ SCCTR       /DONE DELAYING?
0342  5337          JMP .-3

0343  2333          ISZ DLYSC       /YES
0344  5733          JMP I DLYSC     /EXIT
0345  0000  DLYMS,  0
0346  7300          CLA CLL
0347  1023          TAD DELAYM      /GET MS COUNT
0350  3063          DCA MSCTR       /STORE IN MSCTR
0351  1065          TAD MIL1        /GET 1 MS CONSTANT
0352  3064          DCA MILCTR      /STORE IN MILCTR
0353  2064          ISZ MILCTR      /DELAYED 1 MSEC?
0354  5353          JMP .-1
0355  2063          ISZ MSCTR       /DONE DELAYING?
0356  5351          JMP .-5
0357  5745          JMP I DLYMS     /EXIT
```

```
        0400    *. 177+1
                /RANDOM NUMBER GENERATOR SUBROUTINE
  0400  0000    RANGEN, 0
  0401  7200            CLA
  0402  1242            TAD RANTND
  0403  1227            TAD RANDEX
  0404  7640            SZA CLA
  0405  5215            JMP RANTAD
  0406  1231            TAD RANTBL
  0407  3227            DCA RANDEX
  0410  1230            TAD RANCON
  0411  7104            CLL RAL
  0412  7430            SZL
  0413  7001            IAC
  0414  3230            DCA RANCON
  0415  1230    RANTAD, TAD RANCON
  0416  1627            TAD I RANDEX
  0417  3627            DCA I RANDEX
  0420  1243            TAD RANSAV
  0421  7010            RAR
  0422  1627            TAD I RANDEX
  0423  2227            ISZ RANDEX
  0424  3243            DCA RANSAV
  0425  1243            TAD RANSAV
  0426  5600            JMP I RANGEN
  0427  0442    RANDEX, RANTND
  0430  6543    RANCON, 6543
  0431  0432    RANTBL, .+1
  0432  6543            6543
  0433  3210            3210
  0434  0765            0765
  0435  5432            5432
  0436  2107            2107
  0437  7654            7654
  0440  4321            4321
  0441  1076            1076
  0442  7336    RANTND, -.
  0443  0000    RANSAV, 0
```

```
                        /SUBROUTINE TO INITIALIZE BINARY COUNT PATTERN
0444  0000  INITPT, 0
0445  7200        CLA             /SET PT0 = 0
0446  3250        DCA PT0
0447  5644        JMP I INITPT    /EXIT
0450  0000  PT0,  0
0451  0000  PT1,  0
0452  0377  PTMSK, 377

                        /SUBROUTINE TO PROVIDE NEXT BINARY COUNT PATTERN CHARACTER (IN AC)
0453  0000  GETPTT, 0
0454  7200        CLA
0455  1250        TAD PT0         /GET PT0
0456  3251        DCA PT1         /STORE AT PT1
0457  1251        TAD PT1         /GET PT1
0460  7001        IAC             /INCREMENT ACCUMULATOR
0461  0252        AND PTMSK       /LIMIT TO 8 BITS
0462  3250        DCA PT0         /STORE AT PT0
0463  1251        TAD PT1         /GET PT1
0464  5653        JMP I GETPTT    /EXIT

                        /PUNCH/PRINT ONE CHARACTER SUBROUTINE (CHAR IN AC).
0465  0000  PUNCH, 0
0466  2114        ISZ PFLAG       /SET PFLAG
0467  6046        TLS             /PUNCH PRINT
0470  7200        CLA
0471  1114        TAD PFLAG
0472  7640        SZA CLA         /FLAG RESET?
0473  7410        SKP             /NO
0474  5277        JMP .+3         /YES.
0475  6041        TSF             /DONE PRINTING
0476  5271        JMP .-5         /NO.
0477  6042        TCF             /YES, RESET PUNCH/PRINTER FLAG
0500  3114        DCA PFLAG       /RESET PFLAG.
0501  5665        JMP I PUNCH     /EXIT.

                        /SUBROUTINE TO GENERATE RANDOM CHARACTER COUNT, (NOT MORE THAN 77(8))
0502  0000  CHRCNT, 0
0503  4432        JMS I RANDNO    /GO GENERATE RANDOM NUMBER
0504  0317        AND CRMSK       /REMOVE HIGH ORDER 6 BITS
0505  7450        SNA
0506  5303        JMP CHRCNT+1
0507  7041        CIA             /2'S COMPLEMENT IT
0510  3103        DCA SCNT
0511  1702        TAD I CHRCNT
0512  3052        DCA TEMP
0513  1103        TAD SCNT
0514  3452        DCA I TEMP      /STORE AT SPECIFIED ADDRESS
0515  2302        ISZ CHRCNT      /SET UP EXIT
0516  5702        JMP I CHRCNT    /EXIT
0517  0077  CRMSK, 77
```

```
                        /SUBROUTINE TO GENERATE RANDOM DELAY COUNT (NOT MORE THAN 3777(8)).
                        /
0520  0000    DLCNT,
0521  4432              JMS I RANDNO      /GO GENERATE RANDOM NUMBER
0522  0330              AND DLYMSK        /MASK OUT UNDESIRED BITS.
0523  7450              SNA               /ZERO?
0524  5321              JMP DLCNT+1       /YES, GET ANOTHER NUMBER
0525  7041              CIA               /2'S COMPLEMENT IT
0526  3023              DCA DELAYM
0527  5720              JMP I DLCNT       /EXIT
0530  0277    DLYMSK, 277

                        /SUBROUTINE TO COMPARE C(AC) TO CONTENTS STORED AT CALL+1
                        /
0531  0000    CHCK,   0
0532  3345              DCA WCHK          /STORE AC AT WCHK
0533  1731              TAD I CHCK        /GET COMPARE DATA
0534  7041              CIA               /2'S COMPLEMENT IT
0535  1345              TAD WCHK          /ADD C(WCHK)
0536  2331              ISZ CHCK          /SET UP FOR UNEQUAL EXIT
0537  7640              SZA CLA           /EQUAL (AC = 0)
0540  5343              JMP .+3           /NO
0541  2331              ISZ CHCK          /YES,  SET UP FOR EQUAL EXIT
0542  5731              JMP I CHCK        /EQUAL EXIT
0543  1345              TAD WCHK          /RESTORE AC
0544  5731              JMP I CHCK        /UNEQUAL EXIT
0545  0000    WCHK,   0

                        /SYNC ON TAPE SUBROUTINE
                        /
0546  0000    SYNK,   0
0547  4426              JMS I SETCTR      /SET COUNT OF
0550  0566              CTSK              /-256 (DEC) IN
0551  7400              -400              /CTSK
0552  6032    SYNKA,  KCC               /CLEAR AC AND FLAG
0553  6031              KSF               /READY?
0554  5353              JMP .-1           /NO, TEST AGAIN
0555  6034              KRS               /YES, READ
0556  1117              TAD MRBOUT
0557  7640              SZA CLA           /377?
0560  7410              SKP
0561  5746              JMP I SYNK        /YES,  EXIT

0562  2366              ISZ CTSK          /BUMP CHAR CTR +1
0563  5352              JMP SYNKA         /GO READ AGAIN
0564  7402              HLT               /256 CHARS READ,  CAN'T SYNC
0565  5347              JMP SYNK+1        /GO TO SRST
0566  0000    CTSK,   0                 /CHAR COUNTER
```

```
        0600    *  177*1
                /SUBROUTINE TO MOVE VARIABLE LENGTH DATA FIELDS
  0600  0000    MOVE,   0
  0601  7200            CLA
  0602  1600            TAD I MOVE      /GET "FROM ADDR" AND
  0603  3223            DCA FADDR       /STORE AT FADDR
  0604  2200            ISZ MOVE
  0605  1600            TAD I MOVE      /GET "TO ADDR" AND
  0606  3224            DCA TADDR       /STORE AT TADDR,
  0607  2200            ISZ MOVE
  0610  1600            TAD I MOVE      /GET "MOVE COUNT" AND
  0611  3225            DCA MCTR        /STORE AT MCTR,
  0612  2200            ISZ MOVE        /SET UP FOR EXIT.
  0613  7200    MOVEA,  CLA
  0614  1623            TAD I FADDR     /GET "FROM" WORD
  0615  3624            DCA I TADDR     /STORE AT "TO" LOCATION
  0616  2223            ISZ FADDR       /+1 TO "FROM" ADDR
  0617  2224            ISZ TADDR       /+1 TO "TO" ADDR,
  0620  2225            ISZ MCTR        /ALL WORDS MOVED?
  0621  5213            JMP MOVEA       /NO, GO MOVE AGAIN
  0622  5600            JMP I MOVE      /YES, EXIT
  0623  0000    FADDR,  0
  0624  0000    TADDR,  0
  0625  0000    MCTR,   0
```

```
                1000    *, 177+1
                        /PROGRAM NUMBER 3, PUNCHES TEST TAPE WITH 2 CHARACTERS
                        /SPECIFIED IN SYMBOLIC LOCATIONS PTEMP, AND PTEMP1.
1000    7200    PRG3,   CLA
1001    1021            TAD PTEMP       /GET C(PTEMP)
1002    4515            JMS I UPUNCH    /PUNCH C(PTEMP)
1003    1022            TAD PTEMP1      /GET C(PTEMP1)
1004    4515            JMS I UPUNCH    /PUNCH C(PTEMP1)
1005    5200            JMP PRG3        /REPEAT.
                        /
                        /PROGRAM NUMBER 4, PUNCHES TEST TAPE WITH BINARY COUNT PATTERN.
1006    7200    PRG4,   CLA
1007    4507            JMS I INPATT    /INITIALIZE BINARY COUNT PATTERN
1010    4510            JMS I GETPT     /GET BINARY COUNT CHARACTER.
1011    4515            JMS I UPUNCH    /PUNCH CHARACTER
1012    5210            JMP .-2         /REPEAT.
                        /
                        /PROGRAM 5-READS COUNT PATTERN-RANDOM NUMBERED GROUPS,
                        /       FIXED RANDOM DELAY BETWEEN CHARACTERS IN A GROUP.
1013    4506    PRG5,   JMS I SYNC      /SYNC TAPE
1014    3317            DCA ERRCTR      /CLEAR ERROR COUNTER
1015    4507            JMS I INPATT    /INITIALIZE PATTERN.
1016    6032            KCC             /START READER
1017    4513    SRT0A,  JMS I DLYCNT    /GENERATE DELAY COUNT
1020    4512            JMS I CRCNT     /GO GENERATE AND STORE
1021    0101            CTRA            /RANDOM CHAR, COUNT
1022    4510    SRT0B,  JMS I GETPT     /GET PATTERN CHAR,
1023    3276            DCA SBSP        /STORE AT SBSP.
1024    4430            JMS I DLY1MS    /GO DELAY
1025    4271            JMS READCK      /GO READ AND CHECK CHAR,
1026    2101            ISZ CTRA        /GROUP DONE?
1027    5222            JMP SRT0B       /NO,
1030    5217            JMP SRT0A       /YES, START AGAIN
                        /
                        /PROGRAM 6-READS COUNT PATTERN-FIXED DELAY BETWEEN CHARACTERS
1031    4506    PRG6,   JMS I SYNC      /SYNC TAPE
1032    3317            DCA ERRCTR      /CLEAR ERROR COUNTER
1033    4507            JMS I INPATT    /INITIALIZE PATTERN
1034    6032            KCC             /START READER
1035    4510    SRT1A,  JMS I GETPT     /GET PATTERN CHAR,
1036    3276            DCA SBSP        /STORE AT SBSP
1037    4430            JMS I DLY1MS    /GO DELAY
1040    4271            JMS READCK      /GO READ AND CHECK CHAR,
1041    5235            JMP SRT1A       /REPEAT
```

```
                        /PROGRAM 7 - READS CHARS FROM TAPE AND MATCHES AGAINST CHARS
                        /       IN TEMP AND TEMP1, RANDOM DELAY BETWEEN CHARS.
1042  7200      PRG7,   CLA
1043  3317              DCA ERRCTR        /CLEAR ERROR COUNTER
1044  4320              JMS ALIGN         /
1045  6032              KCC               /START READER
1046  4513      SRT2A,  JMS I DLYCNT       /GENERATE DELAY COUNT
1047  4512              JMS I CRCNT       /GO GENERATE AND STORE
1050  0101              CTRA              /RANDOM CHAR COUNT
1051  4350      SRT2B,  JMS GIVE          /GET CHARACTER
1052  3276              DCA SBSP          /STORE AT SBSP
1053  4430              JMS I DLY1MS      /GO DELAY
1054  4271              JMS READCK        /GO READ AND CHECK CHAR
1055  2101              ISZ CTRA          /GROUP DONE?
1056  5251              JMP SRT2B         /NO,
1057  5246              JMP SRT2A         /YES START AGAIN

                        /
                        /PROGRAM 10 - SAME AS SRT2, BUT FIXED DELAY BETWEEN
                        /       CHARS, DELAY IS SPECIFIED IN LOC - DELAYM.
1060  7200      PRG10,  CLA
1061  3317              DCA ERRCTR        /CLEAR ERROR COUNTER.
1062  4320              JMS ALIGN         /
1063  6032              KCC               /START READER
1064  4350      SRT3A,  JMS GIVE          /SET CHARACTER.
1065  3276              DCA SBSP                  /STORE AT SBSP
1066  4430              JMS I DLY1MS      /GO DELAY
1067  4271              JMS READCK        /GO READ AND CHECK CHAR.
1070  5264              JMP SRT3A         /REPEAT.
                        /
1071  0000      READCK, 0
1072  6031              KSF               /READY?
1073  5272              JMP .-1           /TEST AGAIN.
1074  6036              KRB               /READ CLEAR AC AND FLAG.
1075  4511              JMS I CHECK       /GO CHECK CHARACTER WORD.
1076  0000      SBSP,   0                 /
1077  5301              JMP ERRCNT        /ERROR, NO MATCH, GO INC. ERRCNT
1100  5311              JMP HLTTST
1101  2317      ERRCNT, ISZ ERRCTR        /INCREMENT ERROR COUNTER
1102  5305              JMP .+3
1103  7240              CLA CMA           /OFLOW, RESET TO 7777.
1104  3317              DCA ERRCTR
1105  7604              LAS               /READ SR.
1106  0055              AND SR5MSK
1107  7640              SZA CLA           /HALT ON ERROR?
1110  7402              HLT               /YES.
1111  7604      HLTTST, LAS               /READ SR
1112  7700              SMA CLA           /HALT?
1113  5671              JMP I READCK      /NO  EXIT
1114  1317              TAD ERRCTR        /GET ERROR COUNT
1115  7402              HLT               /HALT, ERROR COUNT IN AC
1116  5671              JMP I READCK      /EXIT.
1117  0000      ERRCTR, 0                 /ERROR COUNTER
```

```
                        /
1120  0000  ALIGN,  0
1121  7200           CLA
1122  6031           KSF              /READY?
1123  5322           JMP .-1          /TEST AGAIN,
1124  6034           KRS              /READ CHARACTER,
1125  7041           CIA              /2'S COMPLEMENT IT,
1126  3347           DCA ATEMP        /STORE AT A TEMP,
1127  1347           TAD ATEMP
1130  1021           TAD PTEMP
1131  7650           SNA CLA          /IS IT CHAR IN PTEMP?
1132  5341           JMP AL1          /YES,
1133  1347           TAD ATEMP        /NO,
1134  1022           TAD PTEMP1
1135  7650           SNA CLA          /IS IT CHAR IN PTEMP1?
1136  5344           JMP AL2          /YES,
1137  7402           HLT              /NO, ERROR,
1140  5321           JMP ALIGN+1      /REPEAT,
1141  7040  AL1,     CMA
1142  3346           DCA IND          /SET IND TO -1
1143  5720           JMP I ALIGN
1144  3346  AL2,     DCA IND          /SET IND TO 0,
1145  5720           JMP I ALIGN
1146  0000  IND,     0
1147  0000  ATEMP,   0
1150  0000  GIVE,    0
1151  7200           CLA
1152  2346           ISZ IND          /IS IND = -1?
1153  5357           JMP .+4          /NO,
1154  3346           DCA IND          /YES,
1155  1022           TAD PTEMP1       /GET CHAR FROM TEMP1
1156  5750           JMP I GIVE       /EXIT,
1157  7040           CMA
1160  3346           DCA IND          /SET IND TO -1,
1161  1021           TAD PTEMP        /GET CHAR FROM TEMP,
1162  5750           JMP I GIVE       /EXIT,
                        /
```

```
                1200    *, 177+1
                        /PROGRAM 0, ASR 33/35 TELETYPE BASIC INPUT TESTS,
                        /PROGRAM CHECKS INPUT IOT'S, INTERRUPT, AND READER TIMING
  1200  4426    PRG0,   JMS I SETCTR    /SET KSTART TO INITIAL
  1201  0020            KSTART          /ROUTINE ADDRESS,
  1202  1205            P0TS0
  1203  5604            JMP I .+1       /GO START TEST
  1204  0232            SRSET
                        /
  1205  0000    P0TS0,  0
  1206  1225            P0TS1
                        /ISSUE KCC WITH AC=7777, AC SHOULD GO TO 0,
                        /AC NOT 0 INDICATES KCC FAILURE, TEST IS DONE 1000 TIMES,
  1207  4426            JMS I SETCTR    /SET COUNT OF
  1210  0101            CTRA            /-1000 (DEC) IN
  1211  6030            -1750           /CTRA
  1212  7240            CLA CMA         /SET AC TO 7777
  1213  6032            KCC             /CLEAR AC AND FLAG
  1214  7440            SZA             /IS AC = 0?
  1215  5221            JMP P0E0        /NO, ERROR, GO TO P0E0
  1216  2101            ISZ CTRA        /DONE?
  1217  5212            JMP .-5         /NO, REPEAT
  1220  5424            JMP I CHAIN     /CHAIN
  1221  7402    P0E0,   HLT             /TST0 ERR HALT, KCC DID
                                        /NOT RESULT IN AC = 0
  1222  7240            CLA CMA         /SET A TO 7777
  1223  6032            KCC             /CLEAR AC AND FLAG
  1224  5222            JMP .-2         /REPEAT
  1225  0001    P0TS1,  1
  1226  1246            P0TS2

                        /ISSUE KCC, WAIT 200 MSEC FOR FLAG TO SET,
                        /SKIP ON FLAG, FAILURE TO SKIP INDICATES
                        /THAT FLAG IS NOT SET, OR KSF FAILURE,
                        /TEST IS DONE 100 TIMES,
  1227  4426            JMS I SETCTR    /SET DELAYM
  1230  0023            DELAYM          /TO -200
  1231  7470            -310
  1232  4426    P0TS1A, JMS I SETCTR    /GO SET COUNT OF
  1233  0101            CTRA            /-100 (DEC) IN
  1234  7634            -144            /CTRA
  1235  6032    P0TS1B, KCC             /CLEAR AC AND FLAG
  1236  4430            JMS I DLY1MS    /GO DELAY
  1237  6031            KSF             /SKIP ON FLAG = 1
  1240  5244            JMP P0E1        /ERROR, GO TO E1
  1241  2101            ISZ CTRA        /ALL DONE?
  1242  5235            JMP P0TS1B      /NO, REPEAT
  1243  5424            JMP I CHAIN     /CHAIN
  1244  7402    P0E1,   HLT             /TST1 ERROR HALT, FLAG IS NOT
                                        /SET, OR KSF FAILED
  1245  5232            JMP P0TS1A      /RESTARTING TEST,
```

```
1246  0002  P0IS2,  2
1247  1275          P0TS3
                    /ISSUE KCC, WAIT 200 MSEC FOR FLAG TO BE SET,
                    /SKIP ON FLAG 1000 TIMES TO VERIFY CONSISTENT SKIPPING.
                    /
1250  4426          JMS I SETCTR      /SET DELAYM
1251  0023          DELAYM            /TO -200
1252  7470          -310
1253  4426          JMS I SETCTR      /GO SET COUNT OF 1000
1254  0101          CTRA              /(DEC) IN
1255  6030          -1750             /CTRA
1256  6032  P0TS2A, KCC                /CLEAR AC AND FLAG
1257  4430          JMS I DLY1MS      /GO DELAY
1260  6031          KSF               /SKIP ON FLAG = 1
1261  5267          JMP P0E2A         /DID NOT SKIP, GO TO E2A
1262  6031          KSF               /SKIP ON FLAG = 1
1263  5271          JMP P0E2B         /DID NOT SKIP, GO TO E2B
1264  2101          ISZ CTRA          /ALL DONE?
1265  5262          JMP .-3           /NO,  REPEAT
1266  5424          JMP I CHAIN       /CHAIN
1267  7402  P0E2A,  HLT               /TST2 ERROR HALT, FLAG
                                      /NOT SET OR KSF FAILURE,
1270  5256          JMP P0TS2A
1271  7402  P0E2B,  HLT               /TST2 ERR HALT B,
                                      /KSF FAILURE
1272  6031          KSF               /SKIP ON FLAG = 1
1273  5272          JMP .-1           /REPEAT
1274  5272          JMP .-2           /REPEAT
```

```
1275  0003    P0IS3,  3
1276  1327            P0TS4
                /
                /ISSUE KCC,  WAIT 200 MSECS FOR FLAG TO SET,
                /VERIFY THAT FLAG IS SET,  RESET FLAG (KCC) AND
                /SKIP ON FLAG 500 TIMES TO VERIFY THAT NO
                /SKIP OCCURS WITH FLAG = 0.
                /
1277  4426              JMS I SETCTR      /SET DELATM
1300  0023              DELAYM            /TO -200.
1301  7470              -310
1302  4426              JMS I SETCTR      /SET COUNT OF
1303  0101              CTRA              /-500 (DEC) IN
1304  7014              -764              /CTRA
1305  6032    P0TS3A,   KCC               /CLEAR FLAG
1306  4430              JMS I DLY1MS      /GO DELAY
1307  6031              KSF               /READY?
1310  5320              JMP P0E3A         /NO,  ERROR
1311  6032              KCC               /YES,  RESET FLAG
1312  6031              KSF               /READY?
1313  5315              JMP .+2           /NO,  OK
1314  5322              JMP P0E3B         /YES,  ERROR
1315  2101              ISZ CTRA          /ALL DONE TESTING?
1316  5312              JMP .-4           /NO,  REPEAT
1317  5424              JMP I CHAIN       /YES,  CHAIN
1320  7402    P0E3A,    HLT               /TST3 ERR HALT A,  FLAG
                                          /NOT SET OR KSF FAILURE
1321  5305              JMP P0TS3A        /TRY AGAIN
1322  7402    P0E3B,    HLT               /TST3 ERR HALT B,  FLAG
                                          /FAILED TO RESET, OR KSF
                                          /SKIPPED ERRONEOUSLY.

                /TURN OFF READER BEFORE ENTERING
                /SCOPE LOOP.
1323  6032              KCC               /CLEAR FLAG AND AC
1324  6031              KSF               /SKIP ON FLAG = 1
1325  5323              JMP .-2           /REPEAT
1326  5323              JMP .-3           /REPEAT
```

```
1327  0004    PØTS4,  4
1330  1400            PØTS5
                /THIS ROUTINE CHECKS THAT NO OTHER DEVICE CAN CAUSE AN INTERRUPT,
                /AND THEN CHECKS THAT THE READER FLAG IS CAPABLE OF INTERRUPTING,
1331  4426            JMS I SETCTR    /SET INTERRUPT RETURN
1332  0002            2               /TO PØE4A,
1333  1345            PØE4A
1334  6042    PØTS4A, TCF             /CLEAR PUNCH/PRINTER FLAG
1335  6032            KCC             /CLEAR READER FLAG AFTER
1336  6031            KSF             /IT COMES UP
1337  5336            JMP .-1
1340  6032            KCC             /CLEAR READER FLAG
1341  6001            ION             /ENABLE INTERRUPT
1342  7000            NOP
1343  6002            IOF             /TURN OFF INTERRUPT
1344  5347            JMP .+3
1345  7402    PØE4A,  HLT             /UNEXPECTED INTERRUPT
1346  5334            JMP PØTS4A      /TRY AGAIN
1347  4426            JMS I SETCTR    /SET CTRA TO
1350  0101            CTRA            /-1000
1351  6030            -1750
1352  4426            JMS I SETCTR    /SET INTERRUPT RETURN
1353  0002            2               /TO PØTS4C,
1354  1371            PØTS4C
1355  6032            KCC
1356  6031            KSF             /WAIT FOR READER FLAG
1357  5356            JMP .-1         /TO SET
1360  6001    PØTS4B, ION             /ENABLE INTERRUPT
1361  7000            NOP
1362  7402    PØE4B,  HLT             /READER FLAG FAILED TO INTERRUPT,
                                      /OR INTERRUPT SYSTEM MALFUNCTION
1363  4426            JMS I SETCTR    /SET INTERRUPT RETURN
1364  0002            2               /TO PØTS4C-1,
1365  1370            PØTS4C-1
                /SCOPE LOOP
1366  6001            ION
1367  7000            NOP
1370  5366            JMP .-2
                /
1371  2101    PØTS4C, ISZ CTRA        /DONE?
1372  5360            JMP PØTS4B      /NO,REPEAT
1373  5424            JMP I CHAIN
```

```
      1400      *, 177.1
1400  0005      PØTS5,  5
1401  1424      PØTS6
                /READER TIMING TEST, CHECKS THAT READER FLAG IS =1 NO
                /LATER THAN 110 MILLISECONDS AFTER KCC INSTRUCTION IS ISSUED.
                /
1402  4426              JMS I SETCTR    /SET DELAYM
1403  0023              DELAYM          /TO=-11^0
1404  7622              -156
1405  4426              JMS I SETCTR    /SET COUNT OF
1406  0101              CTRA            /-100 (DEC) IN
1407  7634              -144            /CTRA
1410  6032      PØTS5A, KCC             /START READER, CLEAR PC FLAG
1411  4430              JMS I DLY1MS    /GO DELAY 110 MILLISECS
1412  6031              KSF
1413  5217              JMP PØE5
1414  2101              ISZ CTRA
1415  5210              JMP PØTS5A
1416  5424              JMP I CHAIN
1417  7402      PØE5,   HLT             /TST5 ERR HALT, FLAG NOT=1
                                        /110 MSECS AFTER KCC INSTRUCTION.
1420  6032              KCC             /START READER, CLEAR FLAG, AC.
1421  6031              KSF             /FLAG=1?
1422  5221              JMP .-1         /NO, TEST AGAIN
1423  5220              JMP .-3         /YES, REPEAT.
```

```
1424  0006  PØTS6,  6
1425  7777          7777
                    /READ 256 DIFFERENT CHARACTERS, EACH CHARACTER IS READ 1000 TIMES
                    /TO VERIFY CONSISTENCY OF READING FROM TTI.
                    /
1426  4426          JMS I SETCTR      /SET COUNT OF
1427  0101          CTRA              /-256(DEC)
1430  7400          -400              /IN CTRA
1431  6032  PØTS6A, KCC                /CLEAR AC, FLAG, START RDR'
1432  6031          KSF               /READY?
1433  5232          JMP .-1           /NO, TEST AGAIN,
1434  6034          KRS               /READ CHARACTER,
1435  3266          DCA WTS6A         /SAVE AT WTS6A,
1436  4426          JMS I SETCTR      /SET COUNT OF
1437  0102          CTRB              /CTRB
1440  6030          -1750             /-1000 (DEC) IN
1441  7200  PØTS6B, CLA
1442  6034          KRS               /READ CHARACTER,
1443  3267          DCA WTS6B         /SAVE AT WTS6B
1444  1267          TAD WTS6B         /GET IT BACK,
1445  7041          CIA               /2'S COMPLEMENT IT
1446  1266          TAD WTS6A         /ADD EXPECTED CHAR,
1447  7640          SZA CLA           /RESULT Ø?
1450  5256          JMP PØE6A         /NO, ERROR, GO TO E6A,
1451  2102          ISZ CTRB          /READ CHAR 1000 TIMES?
1452  5241          JMP PØTS6B        /NO, GO READ IT AGAIN,
1453  2101          ISZ CTRA          /YES, READ 256 DIFF, CHARS?
1454  5231          JMP PØTS6A        /NO,
1455  5424          JMP I CHAIN       /YES, CHAIN
1456  1267  PØE6A,  TAD WTS6B
1457  7402          HLT               /TST6 ERR HALT A, AC DISPLAYS
                    /                 /INCORRECTLY READ CHAR, DEPRESS
                    /                 /KEY CONTINUE
1460  7200          CLA
1461  1266          TAD WTS6A
1462  7402  PØE6B,  HLT               /TST6 ERR HALT B, AC DISPLAYS
                    /                 /WHAT THE CORRECT CHAR SHOULD
                    /                 /BE,
1463  7200          CLA
1464  6034          KRS               /READ CHARACTER
1465  5263          JMP .-2           /L
1466  0000  WTS6A,  Ø
1467  0000  WTS6B,  Ø
```

```
                1600    * , 177+1
                        /PROGRAM 1, ASR33/35 TELETYPE BASIC OUTPUT TESTS,
                        /PROGRAM CHECKS OUTPUT IOT,S, INTERRUPT, AND TIMING,
  1600  4426    PRG1,   JMS I SETCTR    /SET KSTART TO INITIAL
  1601  0020            KSTART          /ROUTINE ADDRESS,
  1602  1605            P1TS0
  1603  5604            JMP I ,+1       /GO START TEST
  1604  0232            SRSET
                        /
  1605  0000    P1TS0,  0
  1606  1635            P1TS1
                        /1, TLS AND WAIT 200 MSECS FOR FLAG TO SET, SKIP ON FLAG=1 (TSF),
                        /TSF SHOULD SKIP, OR ERROR HALT P1E0A OCCURS, FLAG NOT SET, OR TSF FAILURE,
                        /2, WITH FLAG=1, SKIP ON FLAG 1000 TIMES TO TEST FOR CONSISTENT SKIPPING,
                        /FAILURE TO SKIP CAUSES ERROR HALT P1E0B,
  1607  4426            JMS I SETCTR    /-200 TO DELAYM
  1610  0023            DELAYM
  1611  7470            -310
  1612  4426            JMS I SETCTR    /-1000 TO CTRA
  1613  0101            CTRA
  1614  6030            -1750
  1615  7200    P1TS0A, CLA             /CLEAR AC
  1616  6046            TLS             /START PRINTER/PUNCH
  1617  4430            JMS I DLY1MS    /DELAY 200 MSECS,
  1620  6041            TSF             /FLAG=1?
  1621  5227            JMP P1E0A       /NO, ERROR
  1622  6041    P1TS0B, TSF             /FLAG=1?
  1623  5231            JMP P1E0B       /ERROR, FAILED TO SKIP,
  1624  2101            ISZ CTRA        /DONE?
  1625  5222            JMP P1TS0B      /NO, REPEAT,
  1626  5424            JMP I CHAIN     /YES, CHAIN
  1627  7602    P1E0A,  HLT CLA         /ERR HALT A, FLAG NOT=1 AFTER
  1630  5215            JMP P1TS0A      /200 MSECS, OR TSF FAILURE
  1631  7602    P1E0B,  HLT CLA         /ERR HALT B, TSF FAILED TO SKIP
  1632  6041            TSF             /SCOPE LOOP, SKIPS ON
  1633  5232            JMP ,-1         /FLAG CONTINUOUSLY,
  1634  5232            JMP ,-2
```

```
1635  0001    P1TS1,  1
1636  1656            P1TS2
                /ISSUE TCF TO CLEAR FLAG, SKIP ON FLAG 1000 TIMES TO VERIFY THAT NO
                /SKIP OCCURS WITH FLAG=0
1637  4426            JMS I SETCTR    /-1000 TO CTRA
1640  0101            CTRA
1641  6030            -1750
1642  6042            TCF             /CLEAR FLAG
1643  6041    P1TS1A, TSF
1644  7410            SKP
1645  5251            JMP P1E1
1646  2101            ISZ CTRA
1647  5243            JMP P1TS1A
1650  5424            JMP I CHAIN
1651  7602    P1E1,   HLT CLA         /ERR HALT, AFTER CLEAR FLAG (TCF),
                                      /TSF INSTRUCTION SKIPPED,
1652  6042            TCF             /SCOPELOOP, CLEARS FLAG,
1653  6041            TSF             /ADN THEN SKIPS ON FLAG
1654  5253            JMP ,-1         /CONTINUOUSLY,
1655  5253            JMP ,-2
                /
1656  0002    P1TS2,  2
1657  1701            P1TS3
                /ISSUE TLS, WAIT FOR FLAG TO SET, CLEAR FLAG (TCF), SKIP ON FLAG=1,NO SKIP
                /SHOULD OCCUR, IF SKIP OCCURS, TCF INSTRUCTION (CLEAR FLAG), FAILED,
1660  4426            JMS I SETCTR    /-100 TO CTRA
1661  0101            CTRA
1662  7634            -144
1663  7200    P1TS2A, CLA
1664  6046            TLS
1665  6041            TSF
1666  5265            JMP ,-1
1667  6042            TCF
1670  6041            TSF
1671  7410            SKP
1672  5276            JMP P1E2
1673  2101            ISZ CTRA
1674  5263            JMP P1TS2A
1675  5424            JMP I CHAIN
1676  7602    P1E2,   HLT CLA         /ERR HALT, TCF FAILED TO RESET
                                      /FLAG,
1677  6042            TCF             /SCOPE LOOP, CLEARS FLAG
1700  5277            JMP ,-1         /CONTINUOUSLY,
```

```
1701  0003    P1TS3,  3
1702  1746            P1TS4
                /THIS ROUTINE CHECKS THAT NO OTHER DEVICE CAN CAUSE AN INTERRUPT,
                /AND THEN CHECKS THAT THE PUNCH/PRINTER FLAG CAN CAUSE AN INTERRUPT.
1703  4426            JMS I SETCTR    /SET INTERRUPT RETURN TO
1704  0002            2               /P1E3A
1705  1717            P1E3A
1706  6032    P1TS3A, KCC             /CLEAR READER FLAG IF UP,
1707  6046            TLS
1710  6041            TSF
1711  5310            JMP .-1
1712  6042            TCF             /CLEAR PUNCH/PRINTER FLAG
1713  6001            ION             /ENABLE INTERRUPTS,
1714  7000            NOP
1715  6002            IOF             /DISABLE INTERRUPTS
1716  5321            JMP .+3
1717  7402    P1E3A,  HLT             /UNEXPECTED INTERRUPT,
1720  5306            JMP P1TS3A      /TRY AGAIN,
1721  4426            JMS I SETCTR    /-1000 TO CTRA,
1722  0101            CTRA
1723  6030            -1750
1724  4426            JMS I SETCTR    /SET INTERRUPT RETURN
1725  0002            2               /TO P1TS3C
1726  1743            P1TS3C
1727  6046            TLS             /START PUNCH/PRINTER
1730  6041            TSF             /FLAG UP?
1731  5330            JMP .-1         /NO, TEST AGAIN
1732  6001    P1TS3B, ION             /YES, ENABLE INTERRUPT
1733  7000            NOP
1734  7402    P1E3B,  HLT             /PRINTER FLAG FAILED TO INTERRUPT
                                      /OR INTERRUPT MALFUNCTION,
1735  4426            JMS I SETCTR    /SET INTERRUPT RETURN
1736  0002            2               /TO P1TS3C-1
1737  1742            P1TS3C-1
1740  6001            ION             /SCOPE LOOP,
1741  7000            NOP
1742  5340            JMP .-2
1743  2101    P1TS3C, ISZ CTRA        /DONE?
1744  5332            JMP P1TS3B      /NO, REPEAT
1745  5424            JMP I CHAIN     /YES, CHAIN
```

```
1746  0004  P1TS4,  4
1747  7777          7777
                    /PUNCH/PRINTER TIMING TEST, CHECKS THAT FLAG IS=1 NO LATER THAN
                    /110 MILLISECONDS AFTER TLS INSTRUCTION
1750  4426          JMS I SETCTR      /-110 TO DELAYM
1751  0023          DELAYM
1752  7622          -156
1753  4426          JMS I SETCTR      /-100 TO CTRA
1754  0101          CTRA
1755  7634          -144
1756  6046  P1TS4A, TLS               /START PUNCH/PRINTER
1757  4430          JMS I DLY1MS      /GO DELAY 110 MSECS,
1760  6041          TSF               /FLAG=1?
1761  5365          JMP P1E4          /NO, ERROR,
1762  2101          ISZ CTRA          /YES, DONE?
1763  5356          JMP P1TS4A        /NO, REPEAT,
1764  5424          JMP I CHAIN       /YES, CHAIN
1765  7602  P1E4,   HLT CLA           /ERR HALT, FLAG NOT 1 110 MSECS
                                      /AFTER TLS INSTRUCTION,
1766  6046          TLS               /SCOPE LOOP, START PRINTER
1767  6041          TSF               /FLAG=1?
1770  5367          JMP .-1           /NO, CHECK AGAIN
1771  5366          JMP .-3           /YES REPEAT,
```

```
        2000    *, 177+1
                /PROGRAM 2, ASR33/35 TELETYPE READER TEST, CHECKS ABILITY OF READER
                /TO CORRECTLY READ AT FULL SPEED AND WITH RANDOM STALLS,
  2000  4426    PRG2,   JMS I   SETCTR  /SET KSTART TO INITIAL
  2001  0020            KSTART          /ROUTINE ADDRESS,
  2002  2005            P2TS0
  2003  5604            JMP I .+1       /GO START TEST
  2004  0232            SRSET
                /
  2005  0000    P2TS0,  0
  2006  2035            P2TS1           /NEXT RTN ADDR,
                /READ 4095 CHARACTERS, AT FULL SPEED, MATCHING EACH CHARACTER
                /READ AGAINST COUNT PATTERN
                /
  2007  4506            JMS I SYNC      /GO SYNC TAPE
  2010  4426            JMS I SETCTR    /SET COUNT OF
  2011  0101            CTRA            /-4095(DEC) IN
  2012  0001            -7777           /CTRA
  2013  6032            KCC             /START READER
  2014  4507            JMS I INPATT    /GO INITIALIZE PATTERN
  2015  4510    P2TS0A, JMS I GETPT     /GET PATTERN CHARACTER
  2016  3223            DCA SB0         /STORE AT SB0
  2017  6031            KSF             /READY?
  2020  5217            JMP .-1         /NO,  TEST AGAIN
  2021  6036            KRB             /YES,  READ CHARACTER
  2022  4511            JMS I CHECK     /GO CHECK FOR CORRECT MATCH
  2023  0000    SB0,    0               /CORRECT CHAR HERE
  2024  5230            JMP P2E0        /ERROR,  GO TO P2E0
  2025  2101    P2T0B,  ISZ CTRA        /OK,  ALL DONE?
  2026  5215            JMP P2TS0A      /NO,  REPEAT
  2027  5424            JMP I CHAIN     /YES,  CHAIN
  2030  7402    P2E0,   HLT             /TST10 ERR HALT,  AC CONTAINS
                                        /CHAR THAT DID NOT MATCH
                                        /AGAINST PATTERN,  EPRESS
                                        /KEY CONTINUE
  2031  7200            CLA
  2032  1223            TAD SB0         /GET CORRECT CHARACTER
  2033  7402            HLT             /AC CONTAINS THE EXPECTED CHARACTER
  2034  5225            JMP P2T0B
```

```
2035  2001    P2TS1,  1
2036  2067            P2TS2
                /READ 2000 CHARACTERS WITH RANDOM DELAY BETWEEN CHARACTERS,
                /MATCH EACH CHARACTER READ AGAINST COUNT PATTERN
                /
2037  4506            JMS I SYNC          /TO SYNC TAPE
2040  4426            JMS I SETCTR        /SET COUNT OF
2041  0101            CTRA                /-2000 (DEC) IN
2042  4060            -3720               /CTRA
2043  6032            KCC                 /START READER
2044  4507            JMS I INPATT        /INITIALIZE PATTERN
2045  4510    P2TS1A, JMS I GETPT         /GET PATTERN CHARACTER
2046  3255            DCA SB1             /STORE AT SB1
2047  4513            JMS I DLYCNT        /GENERATE RANDOM DELAY
2050  4430            JMS I DLY1MS        /DELAY
2051  6031            KSF                 /READY?
2052  5251            JMP .-1             /NO,  TEST AGAIN
2053  6036            KRB                 /YES,  READ CHARACTER
2054  4511            JMS I CHECK         /GO CHECK FOR CORRECT MATCH

2055  0000    SB1,    0                   /CORRECT CHAR HERE
2056  5262            JMP P2E1            /ERROR,  GO TO P2E1
2057  2101    P2T1B,  ISZ CTRA            /OK,  ALL DONE?
2060  5245            JMP P2TS1A          /NO.
2061  5424            JMP I CHAIN         /YES,  CHAIN
2062  7402    P2E1,   HLT                 /TST1 ERR HALT,  AC CONTAINS
                                          /CHARACTER THAT DID NOT MATCH
                                          /AGAINST PATTERN,  DEPRESS
                                          /KEYCONTINUE
2063  7200            CLA
2064  1255            TAD SB1             /GET CORRECT CHARACTER
2065  7402            HLT                 /AC CONTAINS THE EXPECTED
                                          /CHARACTER
2066  5257            JMP P2T1B
```

```
2067  0002    P2TS2,  2
2070  7777            7777
                /READ WITH RANDOM STALL BETWEEN RANDOM CHARACTER GROUPS
                /100 GROUPS READ.
                /
2071  4506            JMS I SYNC          /GO SYNC TAPE
2072  4426            JMS I SETCTR        /SET COUNT OF
2073  0101            CTRA                /-100 (DEC) IN
2074  7634            -144                /CTRA
2075  6032            KCC                 /START READER
2076  4507            JMS I INPATT        /INITIALIZE PATTERN
2077  4513    P2TS2A, JMS I DLYCNT        /SET RANDOM DELAY
2100  4512            JMS I CRCNT         /SET RANDOM CHARACTER
2101  0102            CTRB                /COUNT IN CTRB
2102  4510    P2TS2B, JMS I GETPT         /GET PATTERN CHARACTER
2103  3311            DCA SB2             /AND STORE AT SB2
2104  4430            JMS I DLY1MS        /GO DELAY NO OF
2105  6031            KSF                 /READY?
2106  5305            JMP .-1             /NO, TEST AGAIN
2107  6036            KRB                 /READ CHARACTER
2110  4511            JMS I CHECK         /CHECK FOR CORRECT MATCH
2111  0000    SB2,    0                   /AGAINST SB2 CONTENTS
2112  5320            JMP P2E2            /ERROR, GO TO P2E2
2113  2102            ISZ CTRB            /OK, ALL CHARS FOR GROUP DONE?
2114  5302            JMP P2TS2B          /NO
2115  2101    P2T2C,  ISZ CTRA            /YES, ALL GROUPS DONE?
2116  5277            JMP P2TS2A          /NO
2117  5424            JMP I CHAIN         /YES, CHAIN
2120  7402    P2E2,   HLT                 /TST2 ERROR HALT, AC CONTAINS CHAR THAT
                                          /DID NOT MATCH AGAINST PATTERN, DEPRESS KEY
                                          /CONTINUE
2121  7200            CLA
2122  1311            TAD SB2             /GET CORRECT CHARACTER
2123  7402            HLT                 /AC CONTAINS THE EXPTECTED CHARACTER
2124  5315            JMP P2T2C
                /
```

```
                        /PRG12, ASR33TY AUTOMTIC PUNCH OPTION TEST
                        /CHECKS OPERATION OF AUTOMATIC PUNCH BY PUNCHING RUBOUTS WITH
                        /PCHON, AND ALL Ø'S CHARACTERS WITH PCHOF, THE TAPE RESULTING
                        /SHOULD HAVE ALL RUBOUTS, AS THE TAPE SHOULD NOT ADVANCE
                        /WHEN PUNCHING WITH PCHOFF. THE READER IS USED TO CHECK THE
                        /TAPE, REPEATED FAILURE OF THE PUNCH FEED TO TURN ON WILL
                        /EVENTUALLY BE INDICATED BY TIGHTENING OF SLACK BETWEEN
                        /READER AND PUNCH.
                        /
2244  1242  PRG12,      TAD K7650       /GET (SNA CLA) CODE AND
2245  3326              DCA TTOUTA      /SET AT TTOUTA.
2246  4273              JMS CRPIN       /INITIALIZE DELAY AND CHAR COUNT.
2247  1236              TAD PCHOFF      /TURN OFF PUNCH FEED
2250  4311              JMS TTOUT
2251  1233              TAD RDRON       /TURN ON READER.
2252  4311              JMS TTOUT
2253  4311              JMS TTOUT       /PUNCH ALL Ø'S CHAR.
2254  2102              ISZ CTRB        /DONE FOR N Ø CHARS?
2255  5253              JMP .-2         /NO. REPEAT.
2256  1241              TAD K0377       /YES. PUNCH A RUBOUT
2257  4311              JMS TTOUT
2260  1101              TAD CTRA        /RELOAD CHAR COUNT
2261  3102              DCA CTRB        /INTO CTRB
2262  1235              TAD PCHON       /SET SPFLAG
2263  3237              DCA SPFLAG
2264  1235              TAD PCHON       /TURN ON PUNCH FEED.
2265  4311              JMS TTOUT
2266  1241              TAD K0377       /PUNCH A RUBOUT
2267  4311              JMS TTOUT
2270  2102              ISZ CTRB        /DONE FOR N RUBOUTS?
2271  5266              JMP .-3         /NO. REPEAT.
2272  5246              JMP PRG12+2     /YES. START OVER.
                        /
                        /SUB TO INITIALIZE DELAY CAUSED AND SET CHAR COUNT
2273  0000  CRPIN,      Ø
2274  4426              JMS I SETCTR    /-110 TO DELAYM
2275  0023              DELAYM
2276  7622              -156
2277  4432              JMS I RANDNO    /GET RANDOM NUMBER
2300  0240              AND K0007
2301  7450              SNA             /STILL NON-Ø?
2302  5277              JMP .-3         /NO.
2303  7041              CIA
2304  3101              DCA CTRA        /SET IN CTRA
2305  1101              TAD CTRA        /AND CTRB
2306  3102              DCA CTRB
2307  3237              DCA SPFLAG      /CLEAR SPFLAG
2310  5673              JMP I CRPIN     /EXIT
```

```
                2200            PAGE
                        /PRG11, ASR33Ty AUTOMATIC READER OPTION TEST.
                        /CHECKS THAT READER FLAG RESPONDS TO KCC AFTER "READER ON" COMMAND,
                        /AND THAT FLAG DOES NOT RESPOND TO KCC AFTER "READER OFF" COMMAND,
                        /A TEST IS DONE BETWEEN 1 AND 7 CHARACTERS AFTER EACH READER CONTROL
                        /COMMAND, TEST IS CONTINUOUS RUNNING, FAILURES ARE INDICATED BY HALTS.
                        /
2200  3237      PRG11,  DCA SPFLAG      /CLEAR SPFLAG,
2201  1236              TAD PCHOFF      /TURN OFF PUNCH FEED
2202  4311              JMS TTOUT
2203  4273              JMS CRPIN       /INITIALIZE DELAY AND CHAR COUNT,
2204  1233              TAD RDRON       /TURN ON READER
2205  4311              JMS TTOUT
2206  4430              JMS I DLY1MS    /WAIT AN ADDITIONAL 110 MSECS,
2207  6032              KCC             /ISSUE READER START,
2210  4430              JMS I DLY1MS    /DELAY 110 MSECS,
2211  6031              KSF             /FLAG UP?
2212  7602              HLT CLA         /NO, ERROR, FLAG SHOULD BE UP
2213  2102              ISZ CTRB        /DONE FOR N CHARACTER?
2214  5207              JMP .-5         /NO, REPEAT,
2215  1101              TAD CTRA        /YES, RELOAD CHAR COUNT
2216  3102              DCA CTRB        /INTO CTRB,
2217  1234              TAD RDROFF      /TURN OFF READER
2220  4311              JMS TTOUT
2221  6032              KCC             /CLEAR READER FLAG,
2222  4430              JMS I DLY1MS    /DELAY 110 MSECS
2223  6032              KCC             /ISSUE READER START
2224  4430              JMS I DLY1MS    /WAIT 110 MSECS,
2225  6031              KSF             /FLAG UP?
2226  7410              SKP             /NO, OK,
2227  7602              HLT CLA         /YES, HALT, FLAG SHOULD BE DOWN
2230  2102              ISZ CTRB        /DONE N TIMES
2231  5223              JMP .-6         /NO, REPEAT
2232  5203              JMP PRG11+3     /YES, START OVER,
                        /
2233  0221      RDRON,  221
2234  0223      RDROFF, 223
2235  0222      PCHON,  222
2236  0224      PCHOFF, 224
2237  0000      SPFLAG, 0
2240  0007      K0007,  7
2241  0377      K0377,  377

2242  7650      K7650,  7650
2243  7610      K7610,  7610
```

```
                        /SUB TO OUTPUT DATA TO TELEPRINTER/PUNCH, IF SPFLAG
                        /IS SET, READ A CHARACTER FROM TAPE READER AND CHECK
                        /THAT IT IS A RUBOUT,
2311  0000  TTOUT,  0
2312  6046            TLS             /OUTPUT CHAR,
2313  6041            TSF             /DONE?
2314  5313            JMP .-1         /NO, WAIT
2315  6042            TCF             /YES, CLEAR FLAG,
2316  7200            CLA
2317  1237            TAD SPFLAG
2320  7650            SNA CLA         /PUNCH ON?
2321  5711            JMP I TTOUT     /NO, EXIT,
2322  6032            KCC             /YES, START READER,
2323  6031            KSF             /FLAG UP?
2324  5323            JMP .-1         /NO, WAIT
2325  6034            KRS             /YES, READ CHAR
2326  7650  TTOUTA,    SNA CLA         /OR (SKP CLA)
2327  5711            JMP I TTOUT     /CHAR IS 0, EXIT,
2330  1243            TAD K7610       /GET (SKP CLA) AND
2331  3326            DCA TTOUTA      /SET AT TTOUTA,
2332  6034            KRS             /REREAD CHAR,
2333  1117            TAD MRBOUT      /TAD (-377
2334  7650            SNA CLA         /WAS IT A RUBOUT?
2335  5711            JMP I TTOUT     /YES, OK,
2336  6034            KRS             /NO, ERROR, REREAD CHAR,
2337  7402            HLT             /DISPLAY CHAR,
2340  7200            CLA
2341  5711            JMP I TTOUT     /EXIT,
```

AL1	1141	M40	0074	P1TS3A	1706	RANTND	0442
AL2	1144	MCTR	0625	P1TS3B	1732	RDROFF	2234
ALIGN	1120	MIL1	0065	P1TS3C	1743	RDRON	2233
ATEMP	1147	MILCTR	0064	P1TS4	1746	READCK	1071
C100	0075	MOVE	0600	P1TS4A	1756	RTNNO	0060
C240	0076	MOVEA	0613	P2E0	2030	SB0	2023
CHAIN	0024	MRROUT	0117	P2E1	2062	SB1	2055
CHAINN	0257	MSCTR	0063	P2E2	2120	SB2	2111
CHCK	0531	NXTST	0061	P2T0B	2025	SBSP	1076
CHECK	0111	P0E0	1221	P2T1B	2057	SCCTR	0062
CHRCNT	0502	P0E1	1244	P2T2C	2115	SCNT	0103
CPID	0056	P0E2A	1267	P2TS0	2005	SETCTR	0026
CRCNT	0112	P0E2B	1271	P2TS0A	2015	SHALT	0313
CRMSK	0517	P0E3A	1320	P2TS1	2035	SHLT	0025
CRPIN	2273	P0E3B	1322	P2TS1A	2045	SKIPMA	0077
CTRA	0101	P0E4A	1345	P2TS2	2067	SKIPPA	0100
CTRB	0102	P0E4B	1362	P2TS2A	2077	SPFLAG	2237
CTSK	0566	P0E5	1417	P2TS2B	2102	SR5MSK	0055
CURTST	0057	P0E6A	1456	PCHOFF	2236	SRSET	0232
DELAYM	0023	P0E6B	1462	PCHON	2235	SRST	0031
DLCNT	0520	P0TS0	1205	PFLAG	0114	SRT0A	1017
DLY1MS	0030	P0TS1	1225	PRG0	1200	SRT0B	1022
DLY1SC	0027	P0TS1A	1232	PRG1	1600	SRT1A	1035
DLYCNT	0113	P0TS1B	1235	PRG10	1060	SRT2A	1046
DLYMS	0345	P0TS2	1246	PRG11	2200	SRT2B	1051
DLYMSK	0530	P0TS2A	1256	PRG12	2244	SRT3A	1064
DLYSC	0333	P0TS3	1275	PRG2	2000	START	0200
ERRCNT	1101	P0TS3A	1305	PRG3	1000	STCTR	0322
ERRCTR	1117	P0TS4	1327	PRG4	1006	SYNC	0106
FADDR	0623	P0TS4A	1334	PRG5	1013	SYNK	0546
FLAG	0072	P0TS4B	1360	PRG6	1031	SYNKA	0552
FORWD	0276	P0TS4C	1371	PRG7	1042	TADDR	0624
GETPT	0110	P0TS5	1400	PRGADR	0231	TEMP	0052
GETPTT	0453	P0TS5A	1410	PRGEND	0274	TEMP1	0053
GETRDY	0233	P0TS6	1424	PRGLIM	0034	TEMQ	0070
GIVE	1150	P0TS6A	1431	PRGMSK	0033	TEMR	0071
HLTTST	1111	P0TS6B	1441	PRGNUM	0035	TSTMSK	0054
ID	0215	P1E0A	1627	PRGTAB	0037	TTOUT	2311
INCRTN	0255	P1E0B	1631	PSW	0036	TTOUTA	2326
IND	1146	P1E1	1651	PT0	0450	UMOVE	0116
INITPT	0444	P1E2	1676	PT1	0451	UPUNCH	0115
INPATT	0107	P1E3A	1717	PTEMP	0021	WCHK	0545
K0007	2240	P1E3B	1734	PTEMP1	0022	WTS6A	1466
K0377	2241	P1E4	1765	PTMSK	0452	WTS6B	1467
K100	0104	P1TS0	1605	PUNCH	0465		
K2000	0105	P1TS0A	1615	RANCON	0438		
K7610	2243	P1TS0B	1622	RANDEX	0427		
K7650	2242	P1TS1	1635	RANDNO	0032		
K77	0073	P1TS1A	1643	RANGEN	0400		
KPB	0066	P1TS2	1656	RANSAV	0443		
KPBS	0067	P1TS2A	1663	RANTAD	0415		
KSTART	0020	P1TS3	1701	RANTBL	0431		

ERRORS DETECTED: 0

RUN-TIME: 26 SECONDS

9K CORE USED